
Aufgabe 1 - Einstieg in GPSS
Die Ziel der ersten Aufgabe ist es sich mit der Simulationssprache GPSS und ihrer 
Entwicklungsumgebung vertraut zu machen. Gegeben ist eine informale Beschreibung 
eines einfachen Systems inklusive eines GPSS-Modells und einiger zu beantwortender 
Fragen zum System.

Simulieren Sie das System mit Hilfe des GPSS-Modells in der GPSS-Entwicklungs-
umgebung und beantworten Sie die gestellten Fragen mit Hilfe des automatisch erzeugten 
GPSS-Reports. Verwenden Sie dazu die Studentenversion von GPSS/World, die Sie unter 
der Adresse http://www.minutemansoftware.com beziehen können. Unter dieser Adresse 
ist ebenfalls eine GPSS-Referenz verfügbar, die die Elemente der Sprache erklärt.

Ihre Aufgabe ist die Präsentation des GPSS-Programms in der nächsten Praktikums-
veranstaltung. 

Dazu gehören folgende Punkte:

• Beschreibung des Systems mit Hilfe eines informalen Wort-Bild-Modells zum 
besseren Verständnis für die anderen Studenten

• Zeilenweise Erklärung des GPSS-Programms
• Interpretation der für das Untersuchungsziel relevanten Informationen im GPSS-

Report

Gruppe A: QControl

Informale Systembeschreibung

A component is manufactured by a sequence of three processes, each followed by a short 
two minute inspection. The first process requires 20% of components to be reworked. The 
second and third process require 15% and 5% of components reworked, respectively. 
Sixty percent of components reworked are scrapped and the remaining forty percent need 
reprocessing on the process from which they were rejected.

Manufacturing of a new component commences on average, every 30 minutes, 
exponentially distributed. The time for the first process is given by the following table.

Time For First ProcessTime For First ProcessTime For First ProcessTime For First ProcessTime For First ProcessTime For First ProcessTime For First Process

Frequency .05 .13 .16 .22 .29 .15

Process time
(mins)

10 14 21 32 38 45

The second process takes 15 +/- 6 minutes and the final process time is normally 
distributed with a mean of 24 minutes and a standard deviation of 4 minutes.

1) Simulate the manufacturing process for 100 completed components.
2) Determine the time taken, and the number of components rejected.

http://www.minutemansoftware.com
http://www.minutemansoftware.com


Systemmodell als GPSS-Programm

; GPSS World Sample File - QCONTROL.GPS, by Gerard F. Cummings
***************************************************************** 
*                                                               *
*           Quality Control Program                             *
*           Time units are in minutes                           *
***************************************************************** 
            RMULT        93211      
*  Definitions
Transit     TABLE        M1,100,100,20        ;Transit Time
Process     FUNCTION     RN1,D7
0,0/.05,10/.18,14/.34,21/.56,32/.85,38/1.0,45
*****************************************************************        
            GENERATE     (Exponential(1,0,30))
            ASSIGN       1,FN$Process         ;Process time in P1
Stage1      SEIZE        Machine1
            ADVANCE      P1                   ;Process 1
            RELEASE      Machine1
            ADVANCE      2                    ;Inspection
            TRANSFER     .200,,Rework1        ;20% Need rework
*****************************************************************
Stage2      SEIZE        Machine2
            ADVANCE      15,6                 ;Process 2
            RELEASE      Machine2
            ADVANCE      2                    ;Inspection
            TRANSFER     .150,,Rework2        ;15% Need rework
*****************************************************************
Stage3      SEIZE        Machine3
            ADVANCE      (Normal(1,24,4))     ;Process 3
            RELEASE      Machine3
            ADVANCE      2                    ;Inspection 3
            TRANSFER     .050,,Rework3        ;5% need rework
            TABULATE     Transit              ;Record transit time
            TERMINATE    1
*****************************************************************
Rework1     TRANSFER     .400,,Stage1
            TERMINATE
Rework2     TRANSFER     .400,,Stage2
            TERMINATE
Rework3     TRANSFER     .400,,Stage3
            TERMINATE


